
This gives us an ODSOut object named “t” – we’ll use that 
name to reference methods that build our output. 
 
Again, the DECLARE statement only have to be executed 
once in the data step. 

 provides some powerful, flexible tools for creating reports, like PROC REPORT and PROC 

TABULATE.  With the advent of the Output Delivery System (ODS) you have almost total control over 
how the output from those procedures looks.  But, there are still times where you need (or want) just a 
little more and that’s where the Report Writing Interface (RWI) can help. 
 
The Report Writing Interface is just a fancy way of saying you’re using the ODSOUT object in a data step.  
This object allows you to layout the page, create tables, embed images, add titles and footnotes and 
more – all from within a data step, using whatever data step logic you need.  Also, all the style capabilities 
of ODS are available to you so that your data step created output can have fonts, sizes, colors, 
backgrounds and borders to make your report look just like you want. 
 
This presentation will quickly cover some of the basics of using the ODSOUT object and then walk 
through some of the techniques to create four “real world” examples.  Who knows, you might even go 
home and replace some of your PROC REPORT code – I know I have! 
 
 

THE BASICS OF RWI 
 
PUT statements can still be used in DATA _NULL_ reporting to create reports in a DATA STEP and, with 
all that’s available with ODS styles, they can look very nice (see Lund, 2011 for some discussion of this).  
But, in the new world of DATA _NULL_ reporting tables can be defined right in the data step code and 
even produce many different tables on the same page of output.  Before a discussion of the Report 
Writing Interface (RWI), please understand that this paper is just to get your interest piqued.  This is huge 
topic and you can get much more information in Dan O’Connor’s 2009 SAS Global Forum paper listed in 
the references section.  It contains 40 pages on this topic alone. 
 
In all the example code that follows a couple assumptions are made (unless noted otherwise to expound 
on the example): 

1. They are all within a data step, so no DATA…; or RUN; statements will be shown 
2. Often code that has already been shown and discussed will not be repeated 
3. All of these examples would be creating PDF files, so the “ODS sandwich” statements will be 

shown (ODS PDF file=…; and ODS PDF CLOSE;) 
 
The RWI uses a data step object called ODSOut.  There are “methods” (like functions) of that object that 
will create tables, rows, cells, text, page breaks, lines, etc.  To use an ODSOut object it is first declared 
and given a name – this only has to be done once in the data step and is routinely placed in a conditional 
section of code: 
 
  if _n_ eq 1 then 

    do; 

      declare odsout t(); 

      <other statements> 

    end; 

Once the object is declared you can call “methods” that perform different tasks.  For instance, with our 
object “t,” just a few of the possible methods are: 
 

 t.table_start()  - begins a table  (there is a table_end method that closes a table) 

 t.row_start()  - begins a row in that table – you can have as many rows in the table as you want 
(there is also a row_end method that closes a row) 

 t.format_cell() - inserts a cell (column) into that row – you can have as many cells in a row as 
you want, but each row must have the same number of cells 

 t.format_text() - inserts a line of text (not part of a table) 

 t.line() - puts a horizontal line on the page 

 t.page() - inserts a page break 

 t.title() - creates a page title 
(there is also a footnote method that creates a page footnote) 

 
Note that all of these methods calls have parentheses, which are required – even if empty.  There are 
parameters that can be placed in the parentheses.  For example, the Format_Cell method has a “text” 
parameter that specifies the text to be printed in the cell.  You can specify style attributes in most method 
calls as well, specifying cell borders, appearance of text, line widths, etc., with the OVERRIDES or STYLE 
parameters.  These will be discussed in more detail as we move along. 
 
Here’s a very simple example of using the ODSOUT object.  No dataset will be read by the data step, so 
wrapping the DECLARE statement in an _N_ = 1 loop is not necessary, because the step will execute 
only once. 
 

  data _null_; 

    declare odsout t(); 

 

    t.format_text(text: ‘Hello World); 

  run; 

 
The output from this step would be a single page with the words in the upper left 
corner.  Note that if there are any TITLE or FOOTNOTE statements that are still 
active, these would also appear on the page and the text would be after the 
titles. 
 

CREATING SIMPLE TABLES 
 
But, reports are usually not comprised of just text.  The RWI can define tables of data and there are sets 
of table, row and cell methods that allow us to do that. 
 
 
  t.table_start(); 

    t.row_start(); 

    t.format_cell(text: ‘Hello World); 

    t.row_end(); 

  t.table_end(); 

 
As noted above, the parentheses following the method calls are required, even if no parameters are 
passed.  Also note that the FORMAT_CELL method parameter is exactly the same as that in the 
FORMAT_TEXT method call in the earlier example.  But, the appearance of the output will be very 
different.  The code above, rather than just creating text on the page,  
would create a single cell “table” with the words “Hello World”.  By 
default, the output will be centered on the page, again following any 
titles.  The cell borders, fonts, color, etc. will be determined by the 
ODS STYLE that is being used on the ODS PDF statement. 
 

This method simply writes text to the page.  By 
default is will be left justified and use the font 
attributes associated with the body of the ODS 
style in effect. 

 
 

The creation of a table is pretty straightforward: 

 Start the table 

 Start the first, and only, row 

 Insert a single cell with some text 

 End the row 

 End the table 

It’s simple to make the tables more table-like, by adding more columns and rows.  Getting more columns 
is just a simple matter of having more FORMAT_CELL calls between the ROW_START and ROW_END.  
Note that each cell is bordered individually, just as would be expected in a table.  
 

  t.row_start(); 

    t.format_cell(text: 'hello'); 

    t.format_cell(text: 'world'); 

  t.row_end(); 

 
 
 
Getting more rows is just as simple – add as many ROW_START…ROW_END blocks needed between 
the TABLE_START and TABLE_END. 
 
  t.row_start(); 

    t.format_cell(text: 'hello'); 

    t.format_cell(text: 'world'); 

  t.row_end(); 

  t.row_start(); 

    t.format_cell(text: 'goodbye'); 

    t.format_cell(text: 'earth'); 

  t.row_end(); 

 
Notice above that the columns are automatically sized to fit the largest text in the column in the entire 
table.  To illustrate this further, and to show that in addition to multiple columns and rows, multiple tables 
can also be created in the same data step. 
 
  t.table_start(); 

    t.row_start(); 

      t.format_cell(text: 'hello'); 

      t.format_cell(text: 'world'); 

    t.row_end(); 

  t.table_end(); 

  t.table_start(); 

    t.row_start(); 

      t.format_cell(text: 'goodbye'); 

      t.format_cell(text: 'earth'); 

    t.row_end(); 

  t.table_end(); 

 
 
The two rows are no longer “joined” and widths of the columns are different.  Later, ways to control lots of 
attributes of the tables, which could have made these two tables look much the same, will be shown. 
 
 

DATA-DRIVEN TABLES 
 
It’s not too practical to think of hard-coding all the data to be presented in a table.  Fortunately, in addition 
to a quoted string, the TEXT attribute of the FORMAT_CELL method shown in the examples above can 
take a variable or expression as its value.  This allows for creation of tables from data in datasets or from 
variables created in the data step.  The following examples use the class list from the SASHELP.CLASS 
dataset. 
 

A row can have any number of cells – 
here we’re creating a row with two 
cells (columns), each with a single 
word 
 

A table can have any number of rows 
– here we’ve added a second row to 
the previous example 
 

These are the same two rows we 
had before, but now they are in 
separate tables.  Notice that the 
sizing of the columns is table-
specific and the columns in the two 
rows are no longer the same width 
and that there is now separation 
between the rows. 

  set sashelp.class; 

 

  if _n_ eq 1 then declare odsout t(); 

 

  t.table_start(); 

    t.row_start(); 

      t.format_cell(text: name); 

      t.format_cell(text: height); 

      t.format_cell(text: weight); 

    t.row_end(); 

  t.table_end(); 

 

But, there’s a slight problem with the above code – the TABLE_START and 
TABLE_END methods are going to be called for every iteration of the data step 
and the result is a separate table for every observation, as shown in the table 
snippet to the right.  That might be what is wanted, but probably not.   
 
The solution is simple – place the TABLE_START call in the _N_ eq 1 logic and the TABLE_END call 
with a condition triggered by an END= option on the SET statement. 
 

  set sashelp.class end=done; 

 

  if _n_ eq 1 then 

    do; 

      declare odsout t(); 

 

      t.table_start(); 

      t.row_start(); 

        t.format_cell(text: 'Name'); 

        t.format_cell(text: 'Height (ins)'); 

        t.format_cell(text: 'Weight (lbs)'); 

      t.row_end(); 

    end; 

 

    t.row_start(); 

      t.format_cell(text: name); 

      t.format_cell(text: height); 

      t.format_cell(text: weight); 

    t.row_end(); 

 

  if done then t.table_end(); 

 

Those couple simple changes create a single table, with all the columns the 
same size and the rows joined.  There is a header row to tell the reader 
what’s in the table.  Now is the time to take a look at to control now just what 
appears in the table, but how the table appears. 
 

 

CONTROLLING THE APPEARANCE OF THE TABLES 
 
The programmer has control over almost all aspects of the appearance of the table – text attributes like 
font, color, size and style; cell attributes like borders, size, alignment and background; table attributes like 
spacing and borders.  All can be controlled at most down to the tiniest detail.  There are two parameters 
that can be used in most method calls to do this: OVERRIDES and STYLE.   

Now, bring in a dataset and we’ll use values 
from that to populate our table 
 
Remember, only need to declare the ODSOUT 
object once – do it on the first iteration of the 
data step. 

 

Now, rather than quoted values in the TEXT 
parameter of the FORMAT_CELL method call, 
put the name of a variable.  The contents of that 
variable will be placed in the cell. 

 

The row and cell code is exactly the same as 
before – all that needed to be changed was 
when the table started and stopped.  Now each 
of these rows will be in the same table 

 

Use the END= option to define a variable that 
will be set to 1 (true) when the end of the 
dataset is reached. 

 

In addition to declaring the ODSOUT object, 
move the TABLE_START call to the _N_ eq 1 
block of code.  Also, this is a good place to add 
a single header row to the table.  This 
ROW_START and ROW_END block will only 
be executed once. 

 

When the end of the dataset is reached, end the 
table. 

The real difference between the two parameters is where the list of attributes to apply to the object is 
maintained.  The OVERRIDES parameter lists the “overrides” of the default attributes in the method call 
itself, just like the TEXT parameter.  The STYLE parameter references a style element that is defined in 
PROC TEMPLATE for the ODS STYLE that is currently in use. 
 
A quick example will show how easy, yet powerful, this is.  First, change the appearance of the header 
rows to set them off by overriding a few of the attributes of the cells.     
 
 

 

 

  t.row_start(); 

    t.format_cell(text: 'Name', 

                  overrides: 'background=yellow fontweight=bold cellwidth=30mm'); 

    t.format_cell(text: 'Height (ins)', 

                  overrides: 'background=yellow fontweight=bold cellwidth=25mm'); 

    t.format_cell(text: 'Weight (lbs)', 

                  overrides: 'background=yellow fontweight=bold cellwidth=25mm'); 

  t.row_end(); 

 

 

 

 
Each cell can have its own list of  overrides – here two of them are 
the same across all three cells and one (cellwidth) is not.  In the 
code below, some overrides will also be added to the data rows to 
left-align the name.  Also, the height and weight values are right-
aligned and moved a little over to the left, with the RIGHTMARGIN 
attribute, so that the values are not right against the edge of the cell. 
 

  t.row_start(); 

    t.format_cell(text: name, 

                  overrides: 'just=left'); 

    t.format_cell(text: height, 

                  overrides: 'just=right rightmargin=4mm'); 

    t.format_cell(text: weight, 

                  overrides: 'just=right rightmargin=4mm'); 

  t.row_end(); 

 
As was noted for the TEXT attribute earlier, the OVERRIDE values can be either a quoted string, as 
above, or a character variable (or expression).  In the code above, the height and weight cells could be 
coded as follows, with identical results: 
   

  HW_over = 'just=right rightmargin=4mm'; 

  t.format_cell(text: height, overrides: HW_over); 

  t.format_cell(text: weight, overrides: HW_over); 

 
 
In this simple example, a hard-coded variable is used to set come common attributes.  But, using 
variables instead of hard-coded attribute values also allows a dataset to contain not only the data, but 
information about how the data should be displayed.  In one of the “real world” reports to come, at an 
example of this will be shown. 
 
Another, and probably preferable way, to deal with groups of common attributes is to create an ODS style 
element that contains those attributes.  PROC TEMPLATE is used to create the style, which will then be 
used in the ODS PDF statement that defines the output file. 

The OVERRIDE parameter is used to change 
default attributes.  These are the cells in the 
header row, which will be made bold, with a 
yellow background. 

 

We can use the CELLWIDTH attribute to set the 
width of the columns. 

 

In the first example of this table, all the values 
are centered.  Here, the name is now left-
justified and the other values are right-justified.   

  proc template; 

    define style test; 

    parent=styles.printer; 

 

    style DataCells from body / 

      just=right 

      rightmargin=4mm; 

    end; 

  run; 

 

 

The following FORMAT_CELL calls will again produce the same table as those above with the 
OVERRIDES parameters. 
 

  ods pdf file=<file reference> style=test;  

 

  <... previous data step code ...> 

 

  t.format_cell(text: height, style: 'DataCells');  

  t.format_cell(text: weight, style: 'DataCells'); 

 
 
As might expect be expected, the OVERRIDES and STYLE parameters can also be used together.  If 
both are used, the attribute list is additive, but common attributes use the values in the OVERRIDES.  In 
the code below, the height and weight cells have both parameters. 
 
  t.format_cell(text: height, style: 'DataCells', overrides: 'fontstyle=italic');  

  t.format_cell(text: weight, style: 'DataCells', overrides: 'just=left'); 

 
 
In the example above, attributes set in both places are used.  
The height data is still right justified, with a 4mm margin on 
the right.  But, another attribute has been added – italic text.  
The weight column has one of the attributes set in the 
STYLE changed in the OVERRIDES parameter – the text in 
the cell is now left-justified instead of right justified. 
 
Being able to set attributes in both places gives a lot of control over how the output will look.  Also, there’s 
no real “right” or “wrong” way to do it.  Sometimes, it is very handy to be able to see all the attribute 
values in the data step code, without having to look at the PROC TEMPLATE code.  It is often 
advantageous to see the values where they’re being used.  If this is so, only use the STYLE parameter 
when there are a lot of attributes being set or there are a lot of cells with a common set of attributes. 
 
 

SPANNING COLUMNS AND ROWS  
 
There are header cells in the table above, but information in a table can often be more understandable 
with cells that span multiple columns or rows that contain related information.  There are two parameters 
in the FORMAT_CELL method that control the spanning: COLSPAN, for specifying the number of 
columns the cell should span, and ROWSPAN, for specifying the number of rows that a cell should span. 
 

Create a new ODS style, called TEST, that uses 
the PRINTER style as its base (that’s the 
default style for PDF). 

 
The attributes that were in the OVERRIDES are 
now put in a STYLE element called 
DATACELLS.  The BODY element in the 
PRINTER style is what usually defines the 
attributes of the text in the table.  This overrides 
two of those values. 

 

Use the newly defined style in the ODS PDF 
statement. 

 

Instead of the OVERRIDES parameter, use the 
STYLE parameter.  The style element name is 
in quotes, but could also be a variable that 
contains the style element name. 

A single row added to the table defined above can add set the height and weight columns off a little bit 
from the name column.  This code would immediately precede the row with the height and weight header 
text. 

 
 
  t.row_start(); 

    t.format_cell(); 

    t.format_cell(text: 'Vital Stats',  

                  overrides: 'borderleftcolor=white fontweight=bold', colspan: 2); 

  t.row_end(); 

 
There have to be the same number of columns in each 
row of the table.  There are three columns in the body of 
the table and three in the row above, taking into account 
the vital stats column counts as two.  By default, the cell 
borders would have been drawn around the two merged 
cells, with no line in the middle. 
 
Nothing else new here, except a new attribute: setting the BORDERLEFTCOLOR to white makes the 
appearance a little cleaner.  As expected, the “LEFT” in this attribute could also be “RIGHT,” “TOP,” or 
“BOTTOM.”   
 
Rows can also be spanned with the ROWSPAN parameter.  Suppose that the table above was sorted by 
gender – a far-right column could be added with the gender value and spanned so as not to repeat the 
gender on every row.  A few things must be done to make this work correctly.   
 

 
  by sex; 

 

In the _N_ eq 1 block, add a little to the row above the column headers. 
 

 

  t.table_start(); 

    t.row_start(); 

      t.format_cell(colspan: 2);  

      t.format_cell(text: 'Vital Stats',  

                    overrides: 'borderleftcolor=white fontweight=bold', colspan: 2); 

    t.row_end(); 

 

 

In the data rows, add the gender column.  Note that for the column spanning we know how many columns 
to span and hardcoded the values (2).  But, the number of each type of row is determined by the data.  In 
a prior step, the number of each sex value has been computed and added to each row in a variable called 
“num.”  The cell only needs to be created for the first value of each gender, when the value of 
FIRST.GENDER is true.  This demonstrates another principle – that the method calls themselves can be 
conditional. 
 

 

 

  t.row_start(); 

    if first.sex then t.format_cell(text: put(Sex,$Sex.), 

                                    overrides:'just=left fontweight=bold vjust=top', 

                                    rowspan: num); 

    t.format_cell(text: name,overrides: 'just=left'); 

    t.format_cell(text: height, style: 'DataCells');  

    t.format_cell(text: weight, style: 'DataCells'); 

A blank cell to go over the Name column and 
then “Vital Stats” will be in a cell that spans the 
Height and Weight columns. 

 

The table is sorted by the variable Sex, so that 
the rows can be grouped 

 

The last example had a blank column above the Name.  Now 
we need another blank column above Sex – we could either 
add another FORMAT_CELL or span this over 2 columns 

 

The cell will only be created on the first of a gender value.The 
value of Num determines how many rows will be spanned and 
the VJUST=TOP attribute will move the value to the top of the 
spanned cell.   

  t.row_end(); 

 
The rest of the cells are defined just as they were before, 
with a new spanning row created when the value of sex 
changes. 
 
The complete code for this last example, which covers everything discussed so far, appears in Appendix 
A, along with the final output. 
 
 

LAYING OUT THE PAGE 
 
Those familiar with ODS LAYOUT and ODS REGION statements know that they can be used to place 
output from one or more SAS procedures anywhere on a page.  A powerful feature of the RWI are 
analogous methods for the ODSOUT object.  This means that the tables and text produced in the data 
step can be placed anywhere on the page. 
 
The LAYOUT_ABSOLUTE, REGION and LAYOUT_END method calls allow for tables and text created 
with other method calls to be placed in exact locations on the page.  Like the ODS REGION statement, 
the REGION method call allows for X, Y, HEIGHT and WIDTH parameters defining the position and size 
of the region.  With just a few additional lines of code, the table created above can be placed at a 
designated position on the page.  
 
  if _n_ eq 1 then 

    do; 

      declare odsout t(); 

 

      t.layout_absolute(); 

 

      t.region(x: '2in', y: '2in', height: '8in', width: '5in'); 

 

      <same table start and header row code as before> 

 

    end; 

 

      <same table row code as before> 

 

  if done then  

    do; 

      t.table_end(); 

      t.layout_end(); 

    end; 

 
 
As can be seen in the example to the right, the table has been shifted down and to 
the right.  Note that the Y parameter is sensitive to whatever is already on the page.  
If there are titles or other, no “regionalized” output already on the page, this table 
would begin 2in below that.  (Note: this is the same behavior as the ODS REGION 
statement.) 
 
There is nothing to prevent multiple regions on the same page.  If the class dataset were sorted by 
region, it would be a simple matter to have separate tables for boys and girls, side by side on the page.  
The complete code for this and a snapshot of the output is shown in Appendix B. 
 

The LAYOUT_ABSOLUTE method “turns on” the ability set up 
regions  

The upper-left corner of the table created here will start 2in 
from the left and 2in from the top – the height and width 
parameters are optional, but must be big enough to hold the 
generated output.  If the region is too small, the output is 
suppressed. 

 

After the table is ended, end the layout 

There is also a LAYOUT_GRIDDED method that, along with REGION and LAYOUT_END methods, allow 
for a grid to be defined on the page and output to be placed in one of the grid “cells.”  See Dan 
O’Connor’s paper for examples of this technique. 
 
 
 

REAL WORLD EXAMPLES:  A TABLE OF CONTENTS 
 
Let’s look at some real-life examples that use the RWI to produce the results.  First, a few assumptions 
that will be made for all of the examples: 

 All will be part of a PDF document, so the ODS PDF statement (and corresponding CLOSE) are 
not shown 

 When needed, all use ~ as the ODS ESCAPECHAR 

 For the sake of brevity, often some formatting code will be left out if it does not directly affect the 
current  

 For the sake of clarity, hard-coded values are often shown that, in the production jobs, are 
actually macro calls or macro variable references 

 
 
In this first example, let’s build a table of contents for a report using a data step with just one RWI 
statement.  In this case the information about the pagination of the report is stored in an Excel 
spreadsheet.  This could have been in a SAS dataset, 
database table or any other source that SAS could read.  (We 
kept it in a spreadsheet, so someone else could maintain the 
information.)  The data step reads the spreadsheet and uses 
the “Context Text” and “Page Number” columns as the text in 
the table.   
 
The simple data step below is all that’s needed to create a 
simple table of contents.  The values are used in a 
FORMAT_TEXT method call and written to the file. 
 
 

 

data _null_; 

  set TOC.'Report Contents$'n; 

 

  if _n_ eq 1 then declare odsout rt(); 

 

  LineText = catt(Contents_Text,'~{leaders .}',Page_Number); 

 

  rt.format_text(text: LineText,overrides: "cellwidth=100pct"); 

run; 

 

 
To the right we see the results of the above data 
step.  The text and page numbers are “stretched” 
across the whole page, with the dot leaders 
filling in the gap. 
 
But, as menioned earlier, there are times when 
it’s advantageous to use information in the 
dataset for formatting the results.  Here, we can 
use the Indent_Level column in the spreadsheet to help make the table of contents more readable by 
adding just a few lines of code.   

Use the spreadsheet as the input “dataset” 
 
Declare the ODSOUT object  

 
 
The ODS escape function LEADERS will put 
a line of dots between the Contents _Text  
and Page_Number fields 

 
 Need to tell the string to use 100% of the 
page width – otherwise there would only be 
one dot between the text and page number 

  LineText = catt(Contents_Text,'~{leaders .}',Page_Number); 

 

  if Indent_Level eq 0 then LineText = catt('~{newline}',LineText);  

 

  if Indent_Level eq 0 then LOvr = "cellwidth=100pct"; 

  else LOvr = catx(' ', "cellwidth=100pct",catt('marginleft=',Indent_Level*3,'mm')); 

   

  rt.format_text(text: LineText, overrides: LOvr); 

 
 
 
The new TOC is much easier to read, 
with line breaks between each section 
and subsections indented under the 
main headings.  All of this by using 
information in the data, rather than any 
“fancy” coding.  
 
 
 
 

 
The complete table of contents is shown in Appendix C. 
 

 
REAL WORLD EXAMPLES:  “REPLACING” PROC REPORT 
 
As seen in some of the examples above, RWI can be used to create tables that look a lot like PROCs 
REPORT, PRINT or TABULATE.  Often times, the level of control over the appearance of the tables 
make RWI a great choice.  Also, as shown above, the ability to use data that is not directly displayed in 
the table simplifies the coding that might be required in PROC REPORT, where conditional logic can also 
be use. 
 
In this example, an RWI data step will produce a table that looks very much like something PROC 
REPORT or TABULATE could produce.  The data comes from reports of a statewide survey of students, 
where local results (from a school, district or county) are compared to results of a statewide sample. 
 
In the _N_ eq 1 block, the ODS object “ft” is declared, a table started and the following header row is 
defined.  Notice that a total of eight columns are defined – three with a COLSPAN of 2 and two single-cell 
columns.  The single-cell columns have no text and are very narrow (4mm and 7mm).  They are simply 
used to add a little white space between the columns that contain data. 
 

  ft.row_start(); 

    ft.format_cell(text: '6. How would you describe yourself? (Respondents could....)', 

                   style: 'HeaderRows', 

                   overrides: 'cellwidth=104mm just=left', colspan: 2); 

    ft.format_cell(text: " ",overrides: 'cellwidth=4mm'); 

    ft.format_cell(text: "Your Students",style: 'HeaderRows', 

                   overrides: 'cellwidth=26mm', colspan: 2); 

    ft.format_cell(text: " ",overrides: 'cellwidth=7mm'); 

    ft.format_cell(text: "Statewide",style: 'HeaderRows', 

                   overrides: 'cellwidth=26mm', colspan: 2); 

  ft.row_end(); 

When Indent_Level is 0, add a line break 

 
 

Move the CELLWIDTH to a variable and if 
the INDENT_LEVEL gt 0, add 3mm to the 
left margin for each indent level 

The header (and data rows) are shown in the table below: 

 
The code for the data rows is also similar to code that has been demonstrated earlier.  In this case, there 
are actually eight columns defined, to match those defined in the header row. 

 
  ft.row_start(); 

    ft.format_cell(text: " ",overrides: 'cellwidth=4mm'); 

    ft.format_cell(text: put(VarValue,Race.),overrides: "cellwidth=100mm just=left"); 

    ft.format_cell(text: " "); 

    ft.format_cell(text: put(Percent,NoMissings.),overrides: "cellwidth=12mm"); 

    ft.format_cell(text: put(PlusMinus,PlusMinus.),overrides: "cellwidth=14mm"); 

    ft.format_cell(text: " "); 

    ft.format_cell(text: put(StatePercent,NoMissings.),overrides: "cellwidth=12mm"); 

    ft.format_cell(text: put(StatePlusMinus,PlusMinus.),overrides: "cellwidth=14mm"); 

  ft.row_end(); 

 
Just a couple things to note here: 

1. Notice that the cell widths of columns 1 and 2 (100mm and 4mm) sum to the total in the header 
column (104mm).  The same is true of the columns 4 and 5, and 7 and 8 (12mm and 14mm) 
equaling the 26mm in the header. 

2. The first columns, of width 4mm, is simply used to indent the values.  A LEFTMARGIN attribute of 
4mm in the overrides of the VarValue column, with a total CELLWIDTH of 104mm, would have 
produced the same results.   

 
Now, these results would have been very easy to produce in PROC REPORT.  But, there were some 
circumstances where the desired result would have been much more difficult.  One example was a 
question on honesty in answering the survey.  Surveys from students who responded that they were not 
very honest were removed from the final results, but the complete set of answers to the question were 
required in the report, as shown below. 

 
Having the “Surveys pulled” text spanning the data columns would have been a challenge in SAS 
procedure output.  However, in the data step it is a rather trivial matter.  First, the data row defined above 
is put in an IF…THEN conditional, based on whether a suppression flag is off.  If not, we create a “data 
row” that spans five columns and contains the text that we want.  Notice that five columns are spanned to 
account for the four data columns and the blank, white-space column between them. 

  if not SuppressValues then 

    do; 

      <data row code from above> 

    end; 

  else ft.format_cell(text: 'Surveys pulled',overrides: "just=center", colspan: 5); 

 
 
The real reports that this example comes from are over 55 pages long and contain over 240 tables 
generated using this method, along with other assorted tables and charts.  A couple complete pages from 
a sample report are shown in Appendix D.  Also see the AskHYS information in the reference section for 
links to publicly available versions of these reports. 
 
 

REAL WORLD EXAMPLES:  FREE-FORM REPORTS 
 
In earlier examples, LAYOUT_ABSOLUTE was used to position RWI-generated tables at a particular 
location on the page.  Another use of the layout capabilities of RWI is to “fill out” a form.  This example 
comes from a web-based STD surveillance application.  Data can be entered on line directly from patient 
records or a form can be printed, filled out by hand and the data entered at a later time, if a computer is 
not readily available.  An example of one page of the form is shown in Appendix E-1. 
 
Once the data is in the database, the user can also print the same form, filled with the data that was 
entered into the application.  The RWI is used to put the data onto the page.  There is some tedious initial 
set up, as the X-Y coordinates of all the form fields must be calculated and stored.  But, once it’s done, 
the form can be printed for any case. As with the table of contents example, the information about the 
form fields is stored in a spreadsheet.  Again, this is just for the ease of maintaining the information.   
 
There are two “tricks” to make this work.  The first “trick” is to get the form onto the page.  It would be very 
difficult (never say impossible with SAS) to recreate the form, with all the rounded corners and check 
boxes.  Besides, someone went to a lot of work to create the form, so why not use it as the basis for our 
report.  A simple addition to the style template used to define the page is made in PROC TEMPLATE.  
This is a two-page report and there is a style template for each page – the templates are exactly the 
same, except for the background image.  The images are simply image files of the form pages. 
 
 
  proc template; 

    define style work.myjournal_p1; 

    parent=styles.journal; 

    style body / backgroundimage="<image-location>\STD_Form_1.jpg" 

               margintop=0mm marginleft=0mm height=10.5in width=8in; 

    end; 

 

    define style work.myjournal_p2; 

    parent=styles.journal;  

    style body / backgroundimage="<image-location>\STD_Form_2.jpg" 

               margintop=0mm marginleft=0mm height=10.5in width=8in; 

    end; 

  run; 

 
On the ODS PDF statement, the page 1 style template is used and the image of the form will be the 
background on the page and any output created will be on top of that background image.  
 
  ods pdf file='<file-location>' style=work.myjournal_p1 notoc; 

 
  <code here to generate first page of output> 

The template for page 1 parents off the Journal style and 
modifies the BODY element by adding a 
BACKGROUNDIMAGE attribute 

The template for page 2 is the same as above, but with the 
image of the 2

nd
 page of the form 

Before the code for the 2

nd
 page starts, another ODS PDF statement is used to change the template.  

This will continue to write to the same file defined in the initial ODS PDF statement, but now with the other 
template, a different background image will be used.  This method can be used for as many form pages 
as needed – just change the style templates between each page. 
 
  ods pdf style=work.myjournal_p2; 

 
The second “trick” to make this all work is to get the information about coordinates, fonts and styles into 
the data step.  A HASH object is used to hold all the data in the spreadsheet and it is loaded in the _n_ eq 
1 block, along with the declaration and initial setup of the ODSOUT object. 
 
 

 

if _n_ eq 1 then 

  do; 

    declare hash pl(dataset: "pl.'PageLayout$'n"); 

    pl.defineKey("ItemName"); 

    pl.defineData('Xpos','Ypos','ItemText','FontSize','FontWeight'); 

    pl.defineDone(); 

 

    <more code> 

  end; 

 

The Xpos and Ypos columns hold the position 
on the page where the output will be placed.  
The other columns contain the font size and 
weight, and the value that will be placed at that 
position – either an “X” in a box of the value of 
a variable. 
 
This page of the form displays information 
about the presumptive diagnosis that caused 
the patient to seek treatment. 
 
There are a number of types of information displayed on the form.  A number of fields contain just a single 
value for the case (e.g., patient name, facility name, date of birth).  These are all stored in a dataset 
called PatientInfo.  Other variables can have multiple values for the case, such as the reason for the visit 
(stored in VisitReason).  The database containing information on all cases has already been queried and 
multiple datasets have been created containing the information about the case for which the report is 
being generated.  Multiple SET statements will be used to read all this information, but the data step that 
creates the page will iterate just one time, calling the %DisplayInfo macro the put the information on the 
page. 
 

  set visit.PatientInfo end=endPatient; 

 

  %DisplayInfo(PatientName); 

  %DisplayInfo(PatientID); 

  %DisplayInfo(DOB);  

  %DisplayInfo(DOV); 

  %DisplayInfo(FacilityName); 

  <more fields displayed> 

 

  do while (not endFacType); 

    set visit.VisitReason end=endReason; 

    %DisplayInfo(VisitReasonCode,GetValue=Y); 

  end; 

The information from the spreadsheet is loaded into a HASH 
object called “pl”  - the key field is ItemName and if a search 
on that key is found, all the other data columns are populated. 

No FILE= here, just changing the template 

The single value information is all stored in the dataset 
PatientInfo and the %DisplayInfo macro is called for each 
variable – the variable in the macro call is looked up in the 
spreadsheet and the value will be printed at the location listed 
(e.g., the patient’s name) 

For those variables that can have multiple values, the SET 
statement is placed inside a DO WHILE loop which is 
executed once for each observation in the dataset (could be 
zero).  In these cases, the value of the variable is looked up in 
the spreadsheet and an “X” is placed at the location listed. 

  do while (not endPD); 

    set visit.VisitPD end=endPD; 

    %DisplayInfo(SyndromeCode,GetValue=Y); 

  end; 

 

So, how does that macro work?  The ultimate purpose of the macro is to generate two method calls: 
REGION, to place the text where it’s needed, and FORMAT_TEXT to put the proper value.  The full code 
is listed here. 
 

  %macro DisplayInfo(IN,GetValue=N); 

    %if &GetValue eq N %then %str(ItemName = "&IN";); 

    %else %str(ItemName = vvalue(&IN);); 

     

    rc = pl.find(); 

 

    if not rc then 

      do; 

        vi.region(x: catt(Xpos,'mm'), y: catt(Ypos,'mm')); 

        if missing(FontWeight) then FontWeight = 'medium'; 

        TextOverrides = catx(' ',catt('fontsize=',FontSize,'pt'),catt('fontweight=',FontWeight)); 

        vi.format_text(text: vvaluex(ItemText), overrides: TextOverrides); 

      end; 

  %mend; 

 
If the value of ItemName was found, the XPos and YPos values (from the spreadsheet) are used in the 
REGION method call to set the position on the page where the next output will be placed.  The value of 
ItemText (from the spreadsheet) will be displayed.  Also in the spreadsheet are fontsize and fontweight 
values that are used to build a list of overrides to the default display attributes. 
 
In this example, when PatientName is passed to the macro, it is found in the spreadsheet with the FIND 
method on the hash object and the associated variables are loaded.  So, the X and Y position of the 
region will 39mm and 22.5mm and the patient’s name will print in a 12pt bold font. 
 
In loop that reads the VisitReason reason dataset, the value of the variable VisitReasonCode is looked for 
in the spreadsheet.  A value for symptoms (SYMPT) would print an 8pt bold “X” at 16mm from the left and 
103mm down. 
 
This technique allows for printing of almost any type of form. The complete code to create this page is 
shown in Appendix E-2 and a completed version of the form is shown in Appendix E-3. 
  
 

REAL WORLD EXAMPLES:  REUSING FEATURES! 
 
As seen with the last example, placing RWI code in a macro can often make the actual data step that 
creates the report much shorter and easier to read.  This final example makes heavy use of macros – in 
fact, other than the declaration of the ODSOUT object, there is not a single line of RWI code in the data 
step itself. 
 
The report that is created is for TB case notification for a particular year.  Counts of cases are presented 
in many different ways.  The pages contain headers, tables of different structures and simple rows of 
numbers.  Different macros were created for each type of output.  I won’t present the code here, but will 
describe what some of the macros do and you can envision how “clean” the data step will look. 
 
The SectionHeader macro simply uses a FORMAT_TEXT method to place formatted text at the beginning 
of a section of the report – all that needs to be passed is the text string: 

ItemName is the “key” of the hash – it will either be 
the actual value passed to the macro (“&IN”) or the 
value of a variable name passed to the macro 
(vvalue(&IN)) 

The FIND method on the hash looks for the value in the key 
variable (ItemName) – if it finds it, the return code is 0, and 
all the data fields are filled with the associated values 

Another dataset is looped through and the values 
displayed on the page. 
 
Note that there is no need for the datasets to be in 
any particular order as the X/Y coordinates can 
place the values at any location on the page. 

%SectionHeader(%str(SECTION 2: TB CASE NOTIFICATIONS AND TREATMENT OUTCOMES)) 

 

 
 
 
The SimpleQuestion macro takes three parameters: the question number on the form (just text), the 
question text and the variable that contains the count for that question.  This macro uses code very 
similar to the table of contents example above: 
 
%SimpleQuestion(2.1,New pulmonary smear-positive,Q2_01); 

%SimpleQuestion(2.2,New pulmonary smear-negative,Q2_02); 

%SimpleQuestion(2.3,New pulmonary smear-unknown/not done,Q2_03); 

 

 
This report also mimics a paper-based report that the user can fill out.  For this reason, some of the 
elements on the report are there for compatibility reasons only.  On the paper form, the user has a 
number of places where they can check if data was not available for a section.  The EmptyCellCheckBox 
macro is just a cell in a table with the text shown below – no parameters need to be passed as, in this 
report, the text is always the same. 
 

%EmptyCellCheckBox; 

 

 

 

 
There are a number of tabular presentations of data in this report.  One that is used a number of times is 
a breakdown of different types of cases by age and gender.  The AgeGenderTable macro uses the 
TABLE and ROW methods (START/END) and the FORMAT_CELL methods to create the tables.  The 
question number and text are passed, similar to the SimpleQuestion macro, along with an array 
reference, which contains the numbers for the table.  Notice that we’re taking advantage of the 
BORDER…COLOR attributes to “turn off” the borders of the upper left cell. 
 
%AgeGenderTable(2.16,%str(New pulmonary smear-negative or smear-unknown...,WHO_PSN); 

 

   
There are other table-generation macros that create tables with spanning rows, spanning columns, 
grayed-out cells when data was unavailable, etc. As we’ve seen in the examples so far, there’s not much 
that cannot be done with a little imagination. 
 
The actual report code contains 14 macros that create output types like those shown above. The single 
data step generates a four page report with using over 70 calls to those macros.  Hopefully, this will give 
you the idea that RWI code is perfect for a “modularized” implementation.  There are often reports that 
might use the same type of output over and over and this is so easily done with this method. 
 
The first two pages of this report are shown in Appendix F, showing much of what’s been discussed here.  

CONCLUSION 
 
The Report Writing Interface is a powerful tool in the SAS reporting toolbox.  Even though it’s still “pre-
production” even in v9.3, it has proven to be stable and reliable for many tasks.  Take a look at other 
papers on the SAS Global Forum proceedings web site (see references) and glean what you can.  You’ll 
find that you might often turn to the data step rather than a procedure when it’s time for the next reporting 
task. 

Appendix A 
Example of Basic Table Creation with the Report Writing Interface 

 
proc template; 

  define style test; 

  parent=styles.printer; 

 

  style datacells from body / 

    just=right 

    rightmargin=4mm; 

  style headercells from body / 

    background=yellow fontweight=bold; 

  end; 

run; 

 

proc sql; 

  create table numsex as 

  select sex,count(*) as num 

  from sashelp.class 

  group by 1; 

 

  create table newclass as 

  select c.*, 

         num 

  from sashelp.class c, 

       numsex n 

  where c.sex eq n.sex 

  order by sex,name; 

quit; 

 

proc format; 

  value $Sex  

    'F' = 'Girls' 

    'M' = 'Boys'; 

run; 

 

ods pdf notoc style=test; 

 

data _null_; 

  set newclass end=done; 

  by sex; 

 

  if _n_ eq 1 then 

    do; 

      declare odsout t(); 

 

      t.table_start(); 

        t.row_start(); 

          t.format_cell(colspan: 2); 

          t.format_cell(text: 'Vital Stats',  

                        overrides: 'borderleftcolor=white fontweight=bold', colspan: 2); 

        t.row_end(); 

 

 

      t.row_start(); 

        t.format_cell(text: 'Gender',style: 'HeaderCells', overrides: 'cellwidth=15mm'); 

        t.format_cell(text: 'Name',style: 'HeaderCells', overrides: 'cellwidth=30mm'); 

        t.format_cell(text: 'Height (ins)',style: 'HeaderCells', overrides: 'cellwidth=25mm'); 

        t.format_cell(text: 'Weight (lbs)',style: 'HeaderCells', overrides: 'cellwidth=25mm'); 

      t.row_end(); 

    end; 

 

    t.row_start(); 

      if first.sex then t.format_cell(text: put(Sex,$Sex.), 

                                      overrides:'just=left fontweight=bold vjust=top', rowspan: num); 

      t.format_cell(text: name,overrides: 'just=left'); 

      t.format_cell(text: height, style: 'DataCells');  

      t.format_cell(text: weight, style: 'DataCells'); 

    t.row_end(); 

 

  if done then t.table_end(); 

run; 

 

ods _all_ close; 

Appendix B 
Using the LAYOUT_ABSOLUTE Method for Side-by-Side Output 

 
 

proc template; 

  define style test; 

  parent=styles.printer; 

 

  style datacells from body / 

    just=right 

    rightmargin=4mm; 

  style headercells from body / 

    background=yellow fontweight=bold; 

  end; 

run; 

 

ods _all_ close; 

ods pdf notoc style=test; 

 

data _null_; 

  set sortedclass end=done; 

  by sex; 

 

  retain Group 0; 

 

  if _n_ eq 1 then  

    do; 

      declare odsout t(); 

 

      t.layout_absolute(); 

    end; 

 

  if first.sex then 

    do; 

      t.region(x: catt(Group*4,'in'), y: '1in', width: '3.5in'); 

 

      Group + 1; 

 

      t.table_start(); 

        t.row_start(); 

          t.format_cell(); 

          t.format_cell(text: 'Vital Stats', overrides: 'borderleftcolor=white fontweight=bold', 

                        colspan: 2); 

        t.row_end(); 

 

      t.row_start(); 

        t.format_cell(text: 'Name',style: 'HeaderCells', overrides: 'cellwidth=30mm'); 

        t.format_cell(text: 'Height (ins)',style: 'HeaderCells', overrides: 'cellwidth=25mm'); 

        t.format_cell(text: 'Weight (lbs)',style: 'HeaderCells', overrides: 'cellwidth=25mm'); 

      t.row_end(); 

    end; 

 

    t.row_start(); 

      t.format_cell(text: name,overrides: 'just=left'); 

      t.format_cell(text: height, style: 'DataCells');  

      t.format_cell(text: weight, style: 'DataCells'); 

    t.row_end(); 

 

  if last.sex then t.table_end(); 

  if done then t.layout_end(); 

run; 

 

ods _all_ close; 

Appendix C 
A Simple Table of Contents Created with the Report Writing Interface 

Appendix D 

“Replacing PROC REPORT” with the Report Writing Interface 

Appendix D (cont) 
“Replacing PROC REPORT” with the Report Writing Interface 

Appendix E-1 
Using LAYOUT_ABSOLUTE to Fill a Form 

The Empty Form 

Appendix E-2 
Using LAYOUT_ABSOLUTE to Fill a Form 

Complete Code to Fill the Form 
 
 
proc template; 

  define style work.sti_journal; 

  parent=styles.journal; 

  style TestInfoHeader from document /  

    font_face=Helvetica 

    fontsize=7pt 

    fontweight=bold 

    background=cxFFFFFF  

    just=center; 

  style TestInfo from document /  

    font_face=Helvetica 

    fontsize=7pt 

    just=left; 

  end; 

 

  <page 1 definition> 

 

  define style work.myjournal_p2; 

  parent=work.sti_journal; 

  style body / backgroundimage="&ImagePath\STI_form_FINAL-2.png?width=100%nrstr(%&)height=100%nrstr(%%)" 

   margintop=0mm marginleft=0mm height=10.5in width=8in; 

  end; 

 

  <page 3 definition> 

run; 

 
ods pdf file=<file location> notoc style=work.myjournal_p1; 

 
<page 1 code> 

 
   ods pdf style=work.myjournal_p2; 
 

  data _null_; 

    length ItemName $40 ItemText $100 Xpos Ypos FontSize 4 FontWeight $10 TextOverrides $200; 

    call missing(ItemName, ItemText, Xpos, Ypos, FontSize, FontWeight); 

 

    retain X 'X'; 

 

    if _n_ eq 1 then  

      do; 

        declare hash pl(dataset: "pl.'PageLayout$'n"); 

        pl.defineKey("ItemName"); 

        pl.defineData('Xpos','Ypos','ItemText','FontSize','FontWeight'); 

        pl.defineDone(); 

 

        declare odsout vi(); 

 

        vi.layout_absolute(); 

      end; 

 

    do while (not endPatient); 

      set PatientInfo end=endPatient; 

 

      %DisplayInfo(PatientName_2); 

      %DisplayInfo(PatientID_2); 

      %DisplayInfo(DOB_2); 

      %DisplayInfo(DOV); 

      %DisplayInfo(FacilityName); 

The purpose of the complete code is 
not to be able to reuse it, but to show 
that it’s not all that difficult to create 
a pretty complex form, assuming that 
the data have been processed and 
the position/style information have 
been set up correctly. 
 
This is a technique I use frequently 
and it allows for a SAS-generated 
form that looks exactly as the user 
expects. 

      %DisplayInfo(STIFacilityCode,GetValue=Y); 

      %DisplayInfo(STIClinicCode,GetValue=Y); 

      %DisplayInfo(ClinicOther); 

      %DisplayInfo(PartnerGenderCode,GetValue=Y); 

      %DisplayInfo(PartnersIdentified); 

      %DisplayInfo(PartnersContacted); 

      %DisplayInfo(PartnersTreated); 

      %DisplayInfo(PartnerComments); 

      %DisplayInfo(PatientOutOfJurisdiction); 

      %DisplayInfo(PatientOutOfJurisdictionNoUnk,GetValue=Y); 

    end; 

 

    do while (not endReason); 

      set VisitReason end=endReason; 

      %DisplayInfo(STIVisitReasonCode,GetValue=Y); 

    end; 

 

    do while (not endReasonOther); 

      set VisitReasonOther end=endReasonOther; 

      %DisplayInfo(VisitReasonOther,GetValue=Y); 

    end; 

 

    do while (not endPD); 

      set VisitPD end=endPD; 

      %DisplayInfo(STISyndromeCode,GetValue=Y); 

    end; 

 

    do while (not endTx); 

      set Treatments end=endTx; 

 

      TreatmentCodeMain = compress(STITreatmentCode,'0123456789'); 

      TreatmentCodeByClass = catx('_',TreatmentCodeMain,TreatmentClass); 

      if STITreatmentCode ne TreatmentCodeMain then TreatmentCodeDetail = STITreatmentCode; 

 

      %DisplayInfo(TreatmentCodeByClass,GetValue=Y); 

 

      if STITreatmentCode ne TreatmentCodeMain then 

        do; 

          %DisplayInfo(STITreatmentCode,GetValue=Y); 

        end; 

 

      if TreatmentDate ne . then  

        do; 

          TxDate = put(TreatmentDate,mmddyy10.); 

          DateField = catx('_',TreatmentCodeMain,'Date'); 

          %DisplayInfo(DateField,GetValue=Y); 

        end; 

 

      if STITreatmentSpecify ne '' then  

        do; 

          DetailField = catx('_',TreatmentCodeMain,'Detail'); 

          %DisplayInfo(DetailField,GetValue=Y); 

        end; 

    end; 

 

    %VisitReportInfo; 

 

    vi.layout_end(); 

 

    stop; 

  run; 

 
<page 3 code> 

 

ods pdf close; 

Appendix E-3 
Using LAYOUT_ABSOLUTE to Fill a Form 

The Filled Form 

Appendix F 
Using the Report Writing Interface to Create Complex Output 

Appendix F (cont) 
Using the Report Writing Interface to Create Complex Output 

 
 


